# Copyright 1999-2005 Gentoo Foundation
# Distributed under the terms of the GNU General Public License v2
# $Header: $
# This ebuild is a modified version of the ebuild from http://bugs.gentoo.org/show_bug.cgi?id=102059.
# For new version go to http://gentoo.zugaina.org/
# Small modifications by Ycarus

inherit kde eutils subversion

ESVN_REPO_URI="svn://svn.audioscrobbler.net/player/trunk"

DESCRIPTION="The Last.fm player allows you to listen to their internet radio which is tailored to your music profile"
HOMEPAGE="http://www.last.fm"

LICENSE="BSD"
SLOT="0"
KEYWORDS="~x86 ~amd64"
IUSE="aoss doc"
RESTRICT="nomirror"

DEPEND="=x11-libs/qt-4*"
RDEPEND="=x11-libs/qt-4*
		virtual/x11"

pkg_setup() {
	if ! built_with_use qt gif ; then
		eerror "It appears that $(best_version qt) was compiled with the \"gif\" USE flag disabled."
		eerror
		eerror "In order to use the Last.fm player, you need to enable this USE flag"
		eerror "To do this, run the following in a command window:"
		eerror "echo \"x11-libs/qt gif\" >> /etc/portage/package.use"
		eerror "and recompile qt using \"emerge -avN1 qt\""

		die "no gif support in qt"
	fi
}

src_compile() {
	# Expects /usr/bin/qmake in QT4
	qmake || die "qmake failed"

	emake || die "emake failed"
}

src_install() {
	# The root at which the player, data, and cache
	# are to be installed
	local destination="/opt/lastfm"

	# Install the player
	dodir ${destination}
	exeinto ${destination}
	doexe player

	# Install the icons and pics
	dodir ${destination}/data
	insinto ${destination}/data
	doins data/*

	dodir ${destination}/data/buttons
	insinto ${destination}/data/buttons
	doins data/buttons/*

	# Make a folder such that album art cache works
	diropts -m0775 -g audio
	dodir ${destination}/cache
	keepdir ${destination}/cache

	# make_wrapper is not doing what I want
	exeinto /usr/bin
	local cmd_player="${destination}/player"
	if use aoss ; then
		cmd_player="/usr/bin/aoss ${cmd_player}"
	fi

	local tmpwrapper=$(emktemp)
	cat << EOF > "${tmpwrapper}"
#!/bin/sh
cd "${destination}"
exec ${cmd_player} "\$@"
EOF
	chmod go+rx "${tmpwrapper}"
	newexe "${tmpwrapper}" "lastfm"

	# Install documentation
	if use doc ; then
		dodoc README
	fi

	# Install an icon and make a shortcut for it
	newicon data/icon.png lastfm.png
	make_desktop_entry lastfm "Last.fm Player" lastfm.png
}

pkg_postinst() {
	einfo "In order to use the Last.fm player with your mozilla based browser,"
	einfo "You must follow these steps:"
	einfo
	einfo " 1. Go to \"about:config\" using the location bar"
	einfo " 2. Right-click on the page.  Select \"New\" and then \"String\""
	einfo " 3. Type in the name field: \"network.protocol-handler.app.lastfm\""
	einfo " 4. Type in the value field: \"/usr/bin/lastfm\""
	einfo
	einfo "If you experiance awkward fonts or widgets, you may consider"
	einfo "running \"qtconfig\" and change the settings"
}
